
画面上でモノを動かす方法を学ぼう

X は変 数 です。変数と

は、数字を入れておける

箱のようなものです。

次のプログラムを入力しましょう。

を押して実行すると、●が右に動きます。F5
RUN

10 CLS:CLV

20 X=15:Y=10:A=-1:B=-1:P=10

60 @LOOP

150 LOCATE X,10:PRINT" "

160 X=X+1

170 LOCATE X,10:PRINT"ǩ"

180 WAIT 5

190 GOTO @LOOP

は、エンターキーを押す合図です。

X を � 増やす
プログラム。

X

�

� �

X X

入力してみよう！

スカッシュゲームを
作ろう



Xには最初に15が入っていて、160行が実行されると16になります。

190行のGOTOによって、プログラムは繰り返し実行され、繰り返される

たびにXが1増えます。

LOCATEは、キャラクターを表示する位置を指定する命令です。

変数Xを16、17、18…と増やしながら座標（X,10）の位置に●を表示し

ますから、動いているように見えるというわけです。

160行を次のように変えれば、●が左に動くようになります。
できた

でしょうか？

160 X=X-1

●を左に動かすには、どうしたらよ
いでしょうか？

を押して止めて、 を押して

プログラムリストを表示し、考えてみましょう。

プログラムを変更したら、忘れず を

押してから、 で実行しましょう。

Enter

ESC F4
LIST

F5
RUN

●を左右に動かしてみよう



また を押して止めて、160行を次のように変えてみましょう。ESC

160 X=X+A

もしも、●が画面の左端まで行ったときに、Aに1を

入れることができたら、●が右に進み始めます。

次のプログラムを追加してみましょう。●が画面の

左端で跳ね返ったら成功です！

110 IF X+A=0 A=1

●が画面の右端31の位置まで行った時に、左に跳

ね返るには、どうしたらよいでしょうか？

110行とほとんど同じ、120行を作ってみましょう。

 ？ の部分には、何が入るでしょうか？

120 IF X+A=31 

A= ？

Aもやはり変数で、-1が入っているので●が左に動

きます。

では、Aに1を入れたら、どっちに動くと思いますか？

そう、右です。

-� �

左端まで行き
跳ね返る

右端まで行き
跳ね返る



左右で跳ね返るときに、音が鳴るようにしてみましょう。

●が画面の端で左右に跳ね返り、音が鳴る

ここまでの

プログラム

プログラムを次のように変えましょう。今度は、●が上に動くようになります。

110行と120行に、次のプログラムを付け加え

てください。

を押すのを忘れないように！Enter

:BEEP 40,3

10 CLS:CLV

20 X=15:Y=10:A=-1:B=-1:P=10

60 @LOOP

110 IF X+A=0 A=1:BEEP 40,3

120 IF X+A=31 A=-1:BEEP 40,3

150 LOCATE X,10:PRINT" "

160 X=X+A

170 LOCATE X,10:PRINT"ǩ"

180 WAIT 5

190 GOTO @LOOP

150 LOCATE 15,Y:PRINT" "

160 Y=Y+B

170 LOCATE 15,Y:PRINT"ǩ"

●が上下に動くようにしよう



-05-

そこで、次のプログラムを入力して、上と下でも跳ね返るようにしてみましょう。

130 IF Y+B=0 B=1:BEEP 40,3

140 IF Y+B=22 B=-1:BEEP 40,3

●が斜めに動き、画面の端で跳ね返り、
音が鳴る。

ここまでの

プログラム

●が斜めに動くようにするには、どうしたらよいでしょうか？

10 CLS:CLV

20 X=15:Y=10:A=-1:B=-1:P=10

60 @LOOP

110 IF X+A=0 A=1:BEEP 40,3

120 IF X+A=31 A=-1:BEEP 40,3

130 IF Y+B=0 B=1:BEEP 40,3

140 IF Y+B=22 B=-1:BEEP 40,3

150 LOCATE X,Y:PRINT" "

160 X=X+A:Y=Y+B

170 LOCATE X,Y:PRINT"ǩ"

180 WAIT 5

190 GOTO @LOOP

いくつかのプログラムを追加して、ゲームにして
いきましょう。

次のプログラムを入力しましょう。

DRAW命令で、画面の上と左右に枠を表示します。

画面下で跳ね返るのではなく、パドルで打ち返せるようにしましょう。

140 IF SCR(X+A,Y+B)=61 B=-1:BEEP

15,5:S=S+1

次に、パドルを表示しましょう。

のカーソルキーで、左右に動かせます。← →

30 DRAW 1,1,62,1

40 DRAW 1,1,1,43

50 DRAW 62,1,62,43

70 P=P-BTN(28)+BTN(29)

80 IF P=-1 P=0

90 IF P=25 P=24

100 LOCATE P,22:PRINT" ===== ";

190 IF Y<22 GOTO @LOOP

190行にプログラムを追加します。

●を打ち返せなかったらプログラムが終了します。

2や1にすると、どうなるでしょうか？
速くなる

の？

遅くなるの？

最後に、WAITの5を3にして、●の移動を速くしま

しょう。
180 WAIT 3

SCORE:1

HIT SPACE TO TRY AGAIN!

ハ
イ
ス
コ
ア
目
指
す
ぞ
ー

エ
イ
エ
イ
オ
ー

次のように、ゲームオーバーの処理を追加しましょう。ブーという音が鳴り、得点が表示されます。ス

ペースキーでもう一度ゲームを始めることができるようになります。

200 BEEP60,30

210 LOCATE 5,10:PRINT"SCORE:";S

220 LOCATE 5,12:PRINT"HIT SPACE

TO TRY AGAIN!"

230 IF BTN(32) RUN ELSE CONT

完成したプログラム。
最高得点を狙って遊んでみましょう！

できあがり

10 CLS:CLV

20 X=15:Y=10:A=-1:B=-1:P=10

30 DRAW 1,1,62,1

40 DRAW 1,1,1,43

50 DRAW 62,1,62,43

60 @LOOP

70 P=P-BTN(28)+BTN(29)

80 IF P=-1 P=0

90 IF P=25 P=24

100 LOCATE P,22:PRINT" ===== ";

110 IF X+A=0 A=1:BEEP 40,3

120 IF X+A=31 A=-1:BEEP 40,3

130 IF Y+B=0 B=1:BEEP 40,3

140 IF SCR(X+A,Y+B)=61B=-1:BEEP1

5,5:S=S+1

150 LOCATE X,Y:PRINT" "

160 X=X+A:Y=Y+B

170 LOCATE X,Y:PRINT"ǩ"

180 WAIT 3

190 IF Y<22 GOTO @LOOP

200 BEEP60,30

210 LOCATE 5,10:PRINT"SCORE:";S

220 LOCATE 5,12:PRINT"HIT SP

ACE TO TRY AGAIN!"

230 IF BTN(32) RUN ELSE CONT

遊
ん
で
み
て
ね


